Corrections to the Linvill Normalization Procedure in the NEC Basic Scattering Code

Richard D. Albus

IIT Research Institute

Annapolis, MD, 21401

An error has been found in the Linvill coupling section of the NEC-Basic Scattering Code Version 2.2 (NEC-BSC2.2) [Marhefka, 1982] subroutine OUTPUT. Figure 1 lists the coupling section of subroutine OUTPUT before the error was corrected. On lines 339 and 340 of Figure 1, the short-circuit driving point admittances are incorrectly calculated as the inverse of the open-circuit driving point impedances, i.e., $y_{11} = 1/z_{11}$ and $y_{22} = 1/z_{22}$. The correct expressions for the short-circuit driving point admittances are given in Equation 1.

$$y_{11} = \frac{z_{22}}{z_{11}\ z_{22} - z_{12}\ z_{12}} \qquad (1a)$$

$$y_{22} = \frac{z_{11}}{z_{11}\ z_{22} - z_{12}\ z_{12}} \qquad (1b)$$

However, the conversion from a z-parameter to a y-parameter representation was unnecessary. An important property of consistent two-port systems is that relationships between variables and parameters in any two-port representation will have the same form as the relationships between the corresponding variables, parameters, and terminations of any other two-port representation [Linvill, 1961]. In other words, power ratios will have the same form regardless of whether one uses the z,y,h or g-parameter representations for the two-port model. Therefore, since the Basic

```
C!!!  ANTENNA TO ANTENNA COUPLING REPRESENTATION
      WRITE(6,300)
      WRITE(6,100)
      WRITE(6,100)
      WRITE(6,821)
821   FORMAT(' ANTENNA COUPLING VIA THE REACTION PRINCIPLE'///)
      WRITE(6,150)
      IF(LRECT.AND..NOT.LFQG) WRITE(6,823)
823   FORMAT(1H ,6X,'X',9X,'Y',9X,'Z',9X,'MAGNITUDE',5X,'PHASE',6X
     2,'MAG.**2',8X,'DB')
      IF(.NOT.LRECT.AND..NOT.LFQG) WRITE(6,824)
824   FORMAT(1H ,6X,'R',7X,'THETA',6X,'PHI',8X,'MAGNITUDE',5X
     2,'PHASE',6X,'MAG.**2',8X,'DB')
      IF(LFQG) WRITE(6,825)
825   FORMAT(1H ,15X,'FREQ.',16X,'MAGNITUDE',5X,'PHASE',6X
     2,'MAG.**2',8X,'DB')
      IF(IPRAD.EQ.4) YR11=REAL(1./Z11)
      IF(IPRAD.EQ.4) YR22=REAL(1./Z22)
      IMAX=10*NSN+1
      DO 820 I=NBN,NEN,NSN
      IM=I-1
      CTM=0.5*BABS(CT(I))
      CTP=DPR*BTAN2(AIMAG(CT(I)),REAL(CT(I)))
      GO TO (815,830,835,840),IPRAD
C!!!  UNNORMALIZED REACTION REPRESENTATION
815   CTM2=CTM*CTM
      CTDB=10.*BLOG10(CTM2)
      GO TO 850
C!!!  IMPEDANCE REPRESENTATION
830   Z12=CT(I)/(CI11*CI22)
      CTM2=BABS(Z12)
      CTDB=DPR*BTAN2(AIMAG(Z12),REAL(Z12))
      GO TO 850
C!!!  MODIFIED FRII'S REPRESENTATION
835   CTM2=0.25*CTM*CTM/(PRAD*PRADR)
      CTDB=10.*BLOG10(CTM2)
      GO TO 850

C!!!  LINVILLE REPRESENTATION
840   Z12=CT(I)/(CI11*CI22)
      Y12=Z12/(Z11*Z22-Z12*Z12)
      YY12=Y12*Y12
      YL=BABS(YY12)/(2.*YR11*YR22-REAL(YY12))
      YYL=YL*YL
      CTM2=0.5*YL*(1.+0.25*YYL)
      IF(YL.LT.0.01) GO TO 845
      YYLS=1.-YYL
      CTM2=(1.-SQRT(YYLS))/YL
845   CTDB=10.*BLOG10(CTM2)
850   CONTINUE
      IF(LFQG) GO TO 818
      RXP=RXS+RXI*IM
      TYP=TYS+TYI*IM
      PZP=PZS+PZI*IM
      WRITE(6,810) RXP,TYP,PZP,CTM,CTP,CTM2,CTDB
      GO TO 819
818   FQG=FQGS+FQGI*IM
      WRITE(6,811) FQG,CTM,CTP,CTM2,CTDB
819   IF(I.GT.IMAX) IMAX=IMAX+10*NSN
      IF(I.EQ.IMAX) WRITE(6,400)
820   CONTINUE
      IF(I.NE.IMAX) WRITE(6,400)
      WRITE(6,100)
      WRITE(6,100)
      RETURN
      END
```

Figure 1. The original listing of the antenna coupling section of subroutine NEC-BSC2.2 OUTPUT.

Scattering Code naturally calculates the open-circuit parameters (z-parameters), it will be more convenient to use these parameters to determine the variables associated with the Linvill coupling method.

In addition to the above calculation error, a labeling error exists. On line 354 of Figure 1 the phase angle of the mutual impedance is calculated. However, in the line-printer output, this phase angle is incorrectly labeled as a dB value as indicated by Format statements 824 and 825. Therefore, the line-printer output should be reformatted to correctly label the mutual impedance phase as an angle.

Figure 2 is a listing of the corrected code for the Linvill coupling method using a z-parameter representation for the two-port model as well as enhanced labeling for the line printer output, as used at the DoD Electromagnetic Compatibility Analysis Center (ECAC).

An example is provided to verify the changes to subroutine OUTPUT. This example consists of a horizontal 0.5 wavelength dipole and a horizontal 0.05 wavelength dipole. Both antennas are situated 0.25 wavelengths above an infinite, perfectly conducting ground plane. The NEC-BSC2.2 input data deck is shown in Figure 3. Figure 4 shows graphs of the outputs from the corrected and uncorrected versions of the NEC-BSC2.2. Also shown, for comparison, are results from the NEC-2 program [Burke and Poggio, 1981] for the same problem. Note that the corrected NEC-BSC2.2 results and the NEC-2 results are in close agreement.

```
      IF(LWRITE)WRITE(6,250) RXP,TYP,PZP,PRXR,PTYR,PPZR,PRXI,PTYI,PPZI
      IF (LPDP) THEN
         IF (LPRR) THEN
            WRITE(22,*)DUMMY, XVARAN, PRXR, PTYR, PPZR
         ELSE
            WRITE(22,*)DUMMY, XVARAN, PRXI, PTYI, PPZI
         END IF
      END IF
      GO TO 259
258   FQG=FQGS+FQGI*IM
      IF (LWRITE) WRITE(6,251) FQG,PRXR,PTYR,PPZR,PRXI,PTYI,PPZI
      IF (LPDP) THEN
         IF (LPRR) THEN
            WRITE(22,*) DUMMY, FQG, PRXR, PTYR,PPZR
         ELSE
            WRITE(22,*) DUMMY, FQG, PRXI, PTYI,PPZI
         END IF
      END IF
259   IF(I.GT.IMAX) IMAX=IMAX+10*NSN
      IF (LWRITE) THEN
      IF(I.EQ.IMAX) WRITE(6,400)
      END IF
260   CONTINUE
      IF (LWRITE) THEN
      IF(I.NE.IMAX) WRITE(6,400)
      WRITE(6,100)
      WRITE(6,100)
      END IF
      RETURN
800   CONTINUE
C!!!  ANTENNA TO ANTENNA COUPLING REPRESENTATION
      IF (LWRITE) THEN
         WRITE(6,300)
         WRITE(6,100)
         WRITE(6,100)
         WRITE(6,821)
821   FORMAT(' ANTENNA COUPLING VIA THE REACTION PRINCIPLE'//)
C
C     INITIALIZE THE COLUMN HEADINGS.
C
        MAGVAL = 'MAGNITUDE'
        PHASEV = 'PHASE'
        MAGG2V = '   GAIN   '
        DBVALU = 'GAIN (DB)'
        IF (.NOT. LFQG) THEN
           IF (LRECT) THEN
              VFIRST = 'X'
              VSECND = '  Y  '
              VTHIRD = ' Z '
           ELSE
              VFIRST = 'R'
              VSECND = 'THETA'
              VTHIRD = 'PHI'
           END IF
        END IF
        IF (IPRAD .EQ. 1) THEN
           MAGG2V = ' MAG.**2 '
           DBVALU = '   DB    '
           WRITE(6,1000)
           WRITE(6,150)
```

Figure 2. ECAC's corrected version of the antenna coupling section of NEC-BSC2.2 subroutine OUTPUT.

```
          WRITE(6,2000)
        ELSE IF (IPRAD .EQ. 2) THEN
          MAGG2V = 'MAGNITUDE'
          DBVALU = '  PHASE  '
          WRITE(6,1010)
          WRITE(6,150)
          WRITE(6,2010)
        ELSE IF (IPRAD .EQ. 3) THEN
          WRITE(6,1020)
          WRITE(6,150)
          WRITE(6,2020)
        ELSE
          WRITE(6,1030)
          WRITE(6,150)
          WRITE(6,2020)
        END IF
C
C     WRITE THE HEADINGS FOR EACH COLUMN.
C
        IF (LFQG) THEN
              WRITE(6,2050) MAGVAL, PHASEV, MAGG2V, DBVALU
        ELSE
              WRITE(6,2060) VFIRST, VSECND, VTHIRD, MAGVAL, PHASEV,
     +                       MAGG2V, DBVALU
        END IF
       END IF
       IMAX=10*NSN+1
       DO 320 I=NBN,NEN,NSN
       IM=I-1
       CTM=0.5*BABS(CT(I))
       CTP=DPR*BTAN2(AIMAG(CT(I)),REAL(CT(I)))
       GO TO (815,830,835,840),IPRAD
C!!!   UNNORMALIZED REACTION REPRESENTATION
815    CTM2=CTM*CTM
        CTDB=10.*BLOG10(CTM2)
      GO TO 850
C!!!  IMPEDANCE REPRESENTATION
830   Z12=CT(I)/(CI11*CI22)
      CTM2=BABS(Z12)
      CTDB=DPR*BTAN2(AIMAG(Z12),REAL(Z12))
      GO TO 850
C!!!  MODIFIED FRIIS' REPRESENTATION
835   CTM2=0.25*CTM*CTM/(PRAD*PRADR)
      CTDB=10.*BLOG10(CTM2)
      GO TO 850
C!!!  LINVILL REPRESENTATION
840   Z12=CT(I)/(CI11*CI22)
      ZR11 = REAL( Z11 )
      ZR22 = REAL( Z22 )
      ZZ12=Z12*Z12
      ZL=CABS(ZZ12)/(2.*ZR11*ZR22-REAL(ZZ12))
      ZZL=ZL*ZL
C
      CTM2=0.5*ZL*(1.+0.25*ZZL)
      IF(ZL.LT.0.01) GO TO 845
      ZZLS=1.-ZZL
      CTM2=(1.-SQRT(ZZLS))/ZL
845   CTDB=10.*BLOG10(CTM2)
850   CONTINUE
      IF(LFQG) GO TO 818
      RXP=RXS+RXI*IM
      TYP=TYS+TYI*IM
      PZP=PZS+PZI*IM
      IF ((LPLT) .AND. (.NOT.LFQG)) THEN
```

Figure 2. Continued.

```
            IF(ABS(RXI-TYI).LT.EPSLN) THEN
                IVARAN = 1
            ELSE IF(ABS(TYI-PZI).LT.EPSLN) THEN
                IVARAN = 2
            ELSE
                IVARAN = 3
            END IF
         END IF
         IF (LWRITE) THEN
            WRITE(6,810) RXP,TYP,PZP,CTM,CTP,CTM2,CTDB
         END IF
         IF (LPLT) THEN
             IF (IVARAN .EQ. 1) THEN
                XVARAN = PZP
             ELSE IF (IVARAN .EQ. 2) THEN
                XVARAN = RXP
             ELSE
                XVARAN = TYP
             END IF
             WRITE(19,*)DUMMY,XVARAN,CTDB
         END IF
         GO TO 819
818      FQG=FQGS+FQGI*IM
         IF (LWRITE) THEN
            WRITE(6,811) FQG,CTM,CTP,CTM2,CTDB
         END IF
         IF (LPLT) THEN
            WRITE(19,*) DUMMY,FQG,CTDB
         END IF
819      IF(I.GT.IMAX) IMAX=IMAX+10*NSN
         IF(LWRITE .AND. I.EQ.IMAX) WRITE(6,400)
820      CONTINUE
         IF (LWRITE) THEN
            IF(I.NE.IMAX) WRITE(6,400)
            WRITE(6,100)
            WRITE(6,100)
         END IF
         RETURN
         END
```

Figure 2. Continued.

```
CM:
CM:
CM:
CE: BSC  COUPLING FROM NEC SOURCES
FR:
0.3
US:
1
UN:
1
LP:
T
RG:
0.,0.,.25
90.,0.,90.,90.
-1,.01667,0
.3551E-3,89.984
RG:
-.01667,0.,.25
90.,0.,90.,90.
-1,.01667,0
.1464E-3,89.976
RG:
.01667,0.,.25
90.,0.,90.,90.
-1,.01667,0
.1464E-3,89.976
SG:
-.22222,0.,.25
90.,0.,90.,90.
-1,.05556,0
.2074E-2,-49.546
SG:
-.16667,0.,.25
90.,0.,90.,90.
-1,.05556,0
.4633E-2,-47.518
SG:
-.11111,0.,.25
90.,0.,90.,90.
-1,.05556,0
.6282E-2,-44.915
SG:
-.05556,0.,.25
90.,0.,90.,90.
-1,.05556,0
.7010E-2,-40.997
SG:
.00000,0.,.25
90.,0.,90.,90.
-1,.05556,0
.6863E-2,-35.603
SG:
.05556,0.,.25
90.,0.,90.,90.
-1,.05556,0
.7010E-2,-40.997
SG:
.11111,0.,.25
90.,0.,90.,90.
-1,.05556,0
.6282E-2,-44.915
SG:
.16667,0.,.25
90.,0.,90.,90.
-1,.05556,0
.4633E-2,-47.518
SG:
-.22222,0.,.25
90.,0.,90.,90.
-1,.05556,0
.2074E-2,-49.546
GP:
0
PN:
0.,0.,0.0
0.,0.,90.,0.
T
0.0,0.1,0.25
0.0,0.1,0.0
10
PR:LINVILL
4
(0.55803E-2,-0.39955E-2,(0.10204E-6,0.35507E-3)
(0.11647E+3,0.84823E+2),(0.80936,-0.28163E+4)
XQ:
EN:
```

Figure 3. NEC-BSC2.2 input data.



Figure 4. Comparison of corrected and uncorrected BSC coupling outputs with NEC-2 results.

REFERENCES

Burke, G. T., and Poggio, A. T., "Numerical Electromagnetics Code (NEC) - Method of Moments", NOSC TD 116 (3 parts), Naval Ocean Systems Center, San Diego, CA, Revised 1980,81.

Linvill, J. G., and Gibbons, J. F., "Transistors and Active Circuits," McGraw-Hill Book Co., New York, NY, 1961.

Marhefka, R. J., "Numerical Electromagnetic Code-Basic Scattering Code NEC-BSC(Version 2) Part II: Code Manual," The Ohio State University ElectroScience Laboratory, Columbus, OH, December 1982.